

Amortissement d'un emprunt à annuités ou mensualités constantes

Les formules **VPM**, **INTPER** et **PRINCPER** permettent de programmer un tableau d'emprunt à annuité ou mensualité constantes.

• Fonction : VPM

Cette fonction calcule la mensualité ou l'annuité d'un emprunt.

Annuité, semestrialité, trimestrialité ou mensualité ?

- **Annuité** : saisir le taux d'intérêt annuel (12 %) et saisir un nombre d'annuités (4 pour 4 ans)
- **Semestrialité** : saisir le taux semestriel (6 %) et saisir un nombre de semestre (8 pour 4 ans)
- **Trimestrialité** : saisir le taux trimestriel (3 %) et saisir un nombre de trimestre (16 pour 4 ans)
- **Mensualité** : saisir le taux mensuel (1 %) et saisir un nombre de mensualités (48 pour 4 ans)

• Fonction : INTPER

Syntaxe : **=INTPER(taux;n° période;nbre périodes;Capital)**

Cette fonction calcule l'intérêt des mensualités ou annuités.

• Fonction : PRINCPER

Syntaxe : **=PRINCPER(taux;n° période;nbre périodes;Capital)**

Cette fonction calcule l'amortissement des mensualités ou annuités.

	A	B	C	D	E	F
1	Tableau d'emprunt					
2	Capital :	120 000,00 €				
3	Taux	7,00%				
4	Durée	48				
5	Mensualité	2 873,55 €				
6						
7	Périodes	Capital Début période	Intérêt	Amortissement	Mensualité	Capital fin de période
8	1	120 000,00	700,00	2 173,55	2 873,55	117 826,45
9	2	117 826,45	687,32	2 186,23	2 873,55	115 640,22
10	3	115 640,22	674,57	2 198,98	2 873,55	113 441,24
11	4	113 441,24	661,74	2 211,81	2 873,55	111 229,43
12	5	111 229,43	648,84	2 224,71	2 873,55	109 004,72

	A	B	C	D	E	F
1	Tableau d'emprunt					
2	Capital :	120000				
3	Taux	0,07				
4	Durée	48				
5	Mensualité	=VPM(B3/12;B4;B2)				
6						
7	Périodes	Capital Début période	Intérêt	Amortissement	Mensualité	Capital fin de période
8	1	=B2	=INTPER(\$B\$3/12;A8;\$B\$4;\$B\$2)	=PRINCPER(\$B\$3/12;A8;\$B\$4;\$B\$2)	=D8+C8	=B8-D8
9	2	=F8	=INTPER(\$B\$3/12;A9;\$B\$4;\$B\$2)	=PRINCPER(\$B\$3/12;A9;\$B\$4;\$B\$2)	=D9+C9	=B9-D9
10	3	=F9	=INTPER(\$B\$3/12;A10;\$B\$4;\$B\$2)	=PRINCPER(\$B\$3/12;A10;\$B\$4;\$B\$2)	=D10+C10	=B10-D10
11	4	=F10	=INTPER(\$B\$3/12;A11;\$B\$4;\$B\$2)	=PRINCPER(\$B\$3/12;A11;\$B\$4;\$B\$2)	=D11+C11	=B11-D11
12	5	=F11	=INTPER(\$B\$3/12;A12;\$B\$4;\$B\$2)	=PRINCPER(\$B\$3/12;A12;\$B\$4;\$B\$2)	=D12+C12	=B12-D12